
Overview

This document lists TTCN changes needed to correct a problem in ATS ‘iwd-EUTRA-B2011-03_D11wk15’.

Corrections 

Change 1  –  Correction to calculation of Redundancy Version

Function name
f_EUTRA_SS_SetupSchedulingInfo()

Reason for change
The function that calculates the Subframe Offset and the Redundancy Version for the SIs defines an initial value for the RV, but each iteration that assigns a new RV value overwrites that initial value and it leads to a wrong RV configuration in case of TDD, since all RVs are configured as 0 even for SI3, SI4 and SI5 which are using Subframe Offset 9.

Summary of change
RV configuration in function ‘f_EUTRA_SS_SetupSchedulingInfo()’ is modified to prevent overwriting of the initial value for RV.

TTCN module
EUTRA_ConfigurationSteps.ttcn

MCC160 Comment


Before change:

  function f_EUTRA_SS_SetupSchedulingInfo(  SystemInformationBlockType1 p_SIB1,
                                            EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD )
    return template (value) AllSiSchedul_Type
  {
    var template (value) SiSchedulList_Type v_SiSchedulList;
    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof
    var integer v_SubframeOffset;                          /* offset within the SI-window */
    var integer v_RV := 2;                /* @sic R5s100302 change 9 sic@ */
    var integer I;
    for (I := 0; I < v_NoOfSIs; I:= I + 1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;
         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"
         => there is a maximum of 4 SIs */
     select (p_FDD_TDD) {
        case (FDD) {
          select (I) {
            case (0) {   // Subframe offset for SI1
              v_SubframeOffset := 1;
            }
            case (1) { // Subframe offset for SI2
              v_SubframeOffset := 1;
            }
            // @sic R5-096691: 3 -> 1 sic@
            case (2) { // Subframe offset for SI3
              v_SubframeOffset := 3;
              v_RV := 1;                /* @sic R5s100302 change 9 sic@ */
            }
            case (3) {   // Subframe offset for SI4
              v_SubframeOffset := 7;
              v_RV := 1;                /* @sic R5s100302 change 9 R5s110016 sic@ */
            }
            case (4) {   // Subframe offset for SI5
              v_SubframeOffset := 7;
              v_RV := 1;                /* @sic R5-110471 sic@ */
            }            case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }
          }
        }
       case (TDD) {  // @sic R5-096691 sic@
         select (I) {
           case (0) { // Subframe offset for SI1
             v_SubframeOffset := 4;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 4;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 9;
           }
           case (3) {
             v_SubframeOffset := 9;
           }  // Subframe offset for SI4
           case (4) {
             v_SubframeOffset := 9;    // @sic r5-110471 sic@
           }  // Subframe offset for SI5
           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }
         }
       }
    }
     v_SiSchedulList[I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
     /* @sic R5s100302 change 9 sic@ */
     if (match ( dci_1C, v_SiSchedulList[I].Window[0].DciInfo.Format) ) {
       v_SiSchedulList[I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
     }
    }
    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList);
  }
  

After change:

  function f_EUTRA_SS_SetupSchedulingInfo(  SystemInformationBlockType1 p_SIB1,
                                            EUTRA_FDD_TDD_Mode_Type  p_FDD_TDD )
    return template (value) AllSiSchedul_Type
  {
    var template (value) SiSchedulList_Type v_SiSchedulList;
    var integer v_NoOfSIs := lengthof(p_SIB1.schedulingInfoList);     // TTCN-3 3.4.1: sizeof replaced by lengthof
    var integer v_SubframeOffset;                          /* offset within the SI-window */
    var integer v_RV;
    var integer I;
    for (I := 0; I < v_NoOfSIs; I:= I + 1) {
      /* Assigne the subframe offset according to 36.523-3 Table 7.7.2.2;
         mapping of the SIBs to the SIs is specified in 36.508 cl. 4.4.3.1.2 "Scheduling of system information blocks"
         => there is a maximum of 4 SIs */
     select (p_FDD_TDD) {
        case (FDD) {
          select (I) {
            case (0) {   // Subframe offset for SI1
              v_SubframeOffset := 1;
              v_RV := 2;
            }
            case (1) { // Subframe offset for SI2
              v_SubframeOffset := 1;
              v_RV := 2;
            }
            // @sic R5-096691: 3 -> 1 sic@
            case (2) { // Subframe offset for SI3
              v_SubframeOffset := 3;
              v_RV := 1;                /* @sic R5s100302 change 9 sic@ */
            }
            case (3) {   // Subframe offset for SI4
              v_SubframeOffset := 7;
              v_RV := 1;                /* @sic R5s100302 change 9 R5s110016 sic@ */
            }
            case (4) {   // Subframe offset for SI5
              v_SubframeOffset := 7;
              v_RV := 1;                /* @sic R5-110471 sic@ */
            }            case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }
          }
        }
       case (TDD) {  // @sic R5-096691 sic@
         select (I) {
           case (0) { // Subframe offset for SI1
             v_SubframeOffset := 4;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (1) { // Subframe offset for SI2
             v_SubframeOffset := 4;
             v_RV := 0;    /* @sic R5s110016 sic@ */
           }
           case (2) { // Subframe offset for SI3
             v_SubframeOffset := 9;
             v_RV := 2;
           }
           case (3) {
             v_SubframeOffset := 9;
             v_RV := 2;
           }  // Subframe offset for SI4
           case (4) {
             v_SubframeOffset := 9;    // @sic r5-110471 sic@
             v_RV := 2;
           }  // Subframe offset for SI5
           case else {  FatalError (__FILE__, __LINE__, "No subframe offset defined for that SI"); }
         }
       }
    }
     v_SiSchedulList[I] := cs_SiSchedul_Def(p_SIB1.schedulingInfoList[I].si_Periodicity, v_SubframeOffset);
     /* @sic R5s100302 change 9 sic@ */
     if (match ( dci_1C, v_SiSchedulList[I].Window[0].DciInfo.Format) ) {
       v_SiSchedulList[I].Window[0].DciInfo.RedundancyVersionList[0] := v_RV;
     }
    }
    return cs_SiSchedulList_Def(p_SIB1.si_WindowLength, v_SiSchedulList);
  }

